



Hinweise Zur Anwendung Lerncomputer LC 80 für die Programmierung von EPROM's U2716C

Heft 2

Applikation

Hinweise zur Anwendung des

Lerncomputers LC -80

Autor: Dipl.-Ing. Gunther Zielosko

Programmierung und Löschung von EPROMs U 2716 C

Selbstbau eines Programmiermoduls

Software zum Programmiermodul

Selbstbau eines UV-Löschgerätes

**vob mikroelektronik >karl marx< erfurt
stammbetrieb**

DDR — 5010 Erfurt, Rudolfstraße 47 Telefo 5 80 Telext 061 306



Für die aufgeführten Schaltungen wird keine Gewähr bezüglich Patentfreiheit übernommen.

Nachdruck, auch auszugsweise, nur mit Genehmigung des Herausgebers.

Inhaltsverzeichnis

0.	Einleitung	5
1.	Kurzbeschreibung des EPROMsU 2716 C	6
2.	Vorbereitung des LC-80	8
3.	Die Schaltung des Programmiermoduls	10
4.	Inbetriebnahme des Programmiermoduls	15
5.	Software	16
5.1.	Ladeprogramm im RAM	17
5.1.1.	Hauptprogramm	17
5.1.2.	Unterprogramm INIT	19
5.1.3.	Unterprogramm LESEN	19
5.1.4.	Unterprogramm LEER	20
5.1.5.	Unterprogramm PROGR	20
5.1.6.	Unterprogramm VERGL	22
5.1.7.	Unterprogramm ADVOR	23
5.1.8.	Unterprogramm ADINK	23
5.1.9.	Unterprogramm DATIN	25
5.1.10.	Unterprogramm AUS	25
5.1.11.	Unterprogramm ENDE	26
5.1.12.	Tabelle der Anzeigetexte und Ladeparameter	27
5.2.	Das Ladeprogramm im ROM 3	32
6.	Bedienungsanleitung	32
6.1.	Teiloperationen	33
6.2.	Kombination verschiedener Unterprogramme	35
7.	Löschung von EPROMs	36
7.1.	UV-Löschgerät zum Selbstbau	38
7.2.	Gesundheits-, Arbeits- und Brandschutz	40
8.	Literaturverzeichnis	43

0. Einleitung

Die Elektronikindustrie der DDR stellt den Anwendern der Mikrorechentechnik leistungsfähige Programmier- und Löschtechnik für EPROMs der verschiedenen Generationen zur Verfügung. Dabei handelt es sich entweder um hochproduktive Geräte zur gleichzeitigen Programmierung vieler Schaltkreise (Multiprogrammer) oder um in komplette Systeme eingebaute Programmiermodule.

Beide Konzepte eignen sich nur bedingt für Anwender, die nur selten EPROMs zu programmieren haben oder für Amateure, die ernsthaft in die Mikrorechentechnik einsteigen wollen. Die breite Verfügbarkeit des Lerncomputers LC-80 aus dem VEB Mikroelektronik "Karl Marx" Erfurt - Stammbetrieb gestattet nunmehr auch diesem Anwenderkreis die Nutzung des modernen Speichermediums EPROM.

Die vorliegende Schrift beschreibt ein EPROM-Programmiermodul als Zusatzbaustein für den LC-80, seinen Hardwareaufbau und das dazugehörige Softwaresystem sowie die dazu erforderliche Löschtechnik. Der wichtigste Grundsatz für das vorgestellte System ist die einfache Realisierbarkeit und damit der niedrige Preis. Dadurch bedingt machen sich einige Kompromisse erforderlich, wie z. B. die Beschränkung des Systems auf nur einen EPROM-Typ, den U 2716 C. Dies scheint aber aus zwei Gründen vertretbar, einmal ist der U 2716 C für die nächste Zeit der am meisten verwendete EPROM und zum anderen ist nur dieser Typ im LC-80 einsetzbar. Dennoch ist eine Aufrüstung des Programmiermoduls, insbesondere für den Typ U 2732 C ohne große Probleme möglich.

1. Kurzbeschreibung des EPROMs U 2716 C

Der Schaltkreis U 2716 C ist ein statischer, elektrisch programmierbarer und UV-löschbarer Festwertspeicher (EPROM). Der U 2716 0 wird in n-Kanal-Silicon-Gate-Technologie hergestellt und befindet sich in einem 24poligen DIL-Keramikgehäuse.

Der Schaltkreis besitzt eine Speicherkapazität von 16384 bit mit einer Organisation von 2048 x 8 bit.




Bild 1: Anschlußbelegung und Schaltungskurzzeichen

Bezeichnung der Anschlusses

A 0 ... A 10	Adresseneingänge
D 0 ... D 7	Datenein-/ausgänge
/CE	Chipaktivierungseingang
/OE	Eingang zur Freigabe der Ausgänge
U _{PR}	Programmierungseingang
U _{CC}	Betriebsspannung (+5 V)
U _{SS}	Bezugspotential (0 V)

Der EPROM U 2716 C kann in verschiedenen Betriebsarten arbeiten, die durch die unterschiedliche Beschaltung von drei Anschlüssen ausgewählt werden. Die folgende Tabelle 1 zeigt das, ohne auf die möglichen Toleranzen der einzelnen Pegel einzugehen. Dabei ist L : logisch "0", H . logisch "1" und x ein beliebiger Zustand. Die Pegel sind als Standard-TTL-Pegel zu verstehen.

Funktion	/CE	/OE	U _{PR}	Datenleitungen
Lesen	L	L	+5 V	Ausgänge
Standby	H	x	+5 V	hochohmig
Programmieren	\prod_L^H	H	+25 V	Eingänge
Programmiervergleich	L	L	+25 V	Ausgänge
Programmiersperre	L	H	+25 V	hochohmig

Tabelle 1

In der Betriebsart "Lesen" funktioniert der U 2716 C wie ein ROM. Diese Beschaltung wird im LC-80 benutzt.

Der "Standby"-Zustand inaktiviert den EPROM (z. B. beim Zugriff der CPU auf den RAM-Bereich). Beim U 2716 C wird dabei die Betriebsstromaufnahme um 75 % gesenkt.

Im Gegensatz zu älteren EPROMs kann die Programmierspannung am Anschluß U_{PR} während des gesamten Programmierablaufes statisch anliegen (+25 V ± 1 V). Die Programmierimpulse werden als TTL-Pegel an /CE gegeben, die Zuschaltung der Programmierspannung auf die Speicherzellen erfolgt durch die Innenschaltung des U 2716 C. Alle Speicherstellen eines Bytes (eine Adresse) werden gleichzeitig programmiert, der Programmierimpuls dauert 50 ms.

Die Betriebsart "Programmiervergleich" bietet die Möglichkeit, ohne Abschaltung der Programmierspannung sofort nach dem Programmieren einer Adresse diese zu lesen und auf Richtigkeit zu kontrollieren.

Weitere elektrische Angaben, Grenzwerte und Kennwerte sind den gültigen Datenblättern und der TGL zu entnehmen.

Der EPROM U 2716 C ist UV-löschbar. Durch die spezielle Gehäuseausführung kann die einprogrammierte Information mit UV-Licht gelöscht werden. Zur Löschung werden handelsübliche UV-C-Strahler mit einer Strahlungs-dosis von

$$\sigma_{\min} \leq 15 \text{ Ws/cm}^2 \text{ für Quarzglasdeckel, } \lambda_{\text{UV}} = 254 \text{ nm}$$

$$\sigma_{\min} \leq 30 \text{ Ws/cm}^2 \text{ für Keramikdeckel, } \lambda_{\text{UV}} = 254 \text{ nm}$$

verwendet.

Dabei sollte der Abstand zwischen Gehäuseoberkante des Schaltkreises und dem Lampenkolben $\leq 2,5$ cm betragen. Hinweise zur praktischen Löschtechnologie befinden sich im Abschnitt 7.

2. Vorbereitung des LC-80

Zum effektiven Betrieb des Programmiermoduls sind einige Vorbereitungen am LC-80 erforderlich.

Der Lerncomputer besitzt in der Grundausstattung einen RAM-Arbeitsspeicher von 1 KByte. Dieser reicht nicht aus, um den kompletten Inhalt eines U 2716 C aufzunehmen (2 KByte). Optimal ist daher die Aufrüstung des LC-80-RAM-Speichers um weitere 2 KByte. Bei der Verwendung von 4 CMOS-RAMs U 224 D hält sich der zusätzliche Strombedarf in Grenzen und die Vorschriften zum Betrieb des LC-80 werden eingehalten.

Es werden die vier Positionen oberhalb des in der Grundausstattung vorhandenen RAM-Speichers bestückt, so daß der RAM-Bereich dann von Adresse 2000H bis 2BFFH reicht. Nach dem (sehr sorgfältigen!!) Einlöten der vier RAMs wird deren Funktion überprüft. Neben der RAM-Erweiterung macht es sich erforderlich, wenigstens einen zusätzlichen ROM-Steckplatz zu schaffen, um programmierte EPROMs im LC-80 auch einsetzen zu können. Auch hier ist zu beachten, daß die Stromversorgung des LC-80 (und damit die Temperatur des integrierten Festspannungsreglers sowie der Gleichrichterdioden) weiter belastet wird. Praktische Versuche haben jedoch gezeigt, daß funktionell keine Störungen zu erwarten sind.

Es ist eine 24polige Steckfassung (VEB KSG) in die mittlere ROM-Position (ROM 3) einzulöten. Diese Position hat die Adressen 1000 H bis 17FFH.

Abschließend muß noch eine Drahtbrücke oberhalb des USER-Steckverbinders X 1 (+5 V an X 1/A 1) eingelötet werden, um die Betriebsstromversorgung des Programmiermoduls zu sichern.

Alle Zusatzbauelemente sind mit Abblock-Kondensatoren - 10 nF zu beschalten. Lötaugen dafür sind an allen ROM- und RAM-Positionen des LC-80 vorgesehen.

Im Anschluß an diese Hardware-Änderungen sollte eine komplette Überprüfung aller zusätzlich eingebauten Bauelemente erfolgen. Dies kann elegant mittels kurzer Prüfprogramme erfolgen, die alle Speicherplätze im Zusatz-RAM beschreiben und lesen. Die ROM-Position 3 kann dazu mit einem programmierten U 2716 C oder U 505 D bestückt und geprüft werden.

3. Die Schaltung des Programmiermoduls

Bild 2 zeigt die komplette Schaltung des Programmiermoduls. Das Modul ist über einen 26poligen Steckverbinder (gehört zum Lieferumfang des LC-80) mit dem USBR-PORT des Lerncomputers verbunden. Alle Daten- und Steuersignale liefert die USER-PIO. Port A wird multivalent genutzt:

- zum Datentransport im Direktverkehr mit dem U 2716 C
- zur Adressierung des U 2716 C über die Zwischenspeicher V 4042 D (IC 1 ... IC 3).

Port 3 organisiert die komplette Systemsteuerung. Dabei sind

- B 0 und B 1 für die Auswahl von Adreßbereichen oder Daten,
- B 2 für die Auswahl des Zustandes von /CE und
- B 3 für die Auswahl des Zustandes vom /OE und die Zuschaltung der Programmierspannung an U_{PR}

verantwortlich

Die Tabelle 2 zeigt die Zuordnung von Schaltfunktionen auf die jeweiligen Signale von B 0 und B 1.

B 0	B 1	Wirkung
0	0	U_{CC} für U 2716 C ist abgeschaltet
1	0	Port A liefert Adressen A 8 ... A 10, U_{CC} liegt an +5 V
0	1	Port A liefert Adressen A 0 ... A 7, U_{CC} liegt an +5 V
1	1	Port A dient zum Datentransfer, U_{CC} liegt an +5 V

Tabelle 2

Mit den logischen Verknüpfungen der Signale B 0 und B 1 wird die Übernahme der folgenden Ausgabe des Port A vorbereitet.

Die Signale ARDY und /ASTB bilden dann den Übernahmeimpuls für die drei Adreßspeicher IC 1 ... IC 3.

B 2 steuert direkt den /CE-Eingang und B 3 den /OE-Eingang des U 2716 C.

Über den Zustand /OE wird außerdem die Programmierspannung an U_{PR} des U 2716 C gesteuert ($/OE = B 3 = H \rightarrow U_{PR} +25 V$). Ist $/OE = B 3 = L$, sperrt T 3 und die Spannung an U_{PR} wird über D 1 vom Zustand an U_{CC} bestimmt. Liegen dort +5 V an, erhält U_{PR} den Pegel von ca. +4,6 V (Germaniumdiode 1).

Auch U_{CC} ist softwaremäßig steuerbar, wie Tabelle 2 gezeigt hat. Wenn B 0 und B 1 L-Pegel führen, sperrt T 1 und schaltet damit die Betriebsspannung für den EPROM ab. Mit den gezeigten logischen Verknüpfungen können nun alle benötigten Funktionen des U 2716 C per Software aufgerufen werden. Die Schaltung wurde so ausgelegt, daß vor Beginn und nach Beendigung der Operationen am U 2716 C alle Signale und Betriebsspannungen abgeschaltet werden können.

Noch einige Hinweise zur Bereitstellung der Betriebsspannungen. Die +5 V-Versorgung erfolgt durch den LC-80. Auch dies ist eine Zusatzbelastung des internen Netzteiles des Lerncomputers. Funktionell sind keine Störungen zu befürchten, da der integrierte Festspannungsregler bis 1 A und mehr belastbar ist. Auch die zusätzliche Erwärmung ist vertretbar, da der EPROM immer nur kurzzeitig betrieben wird. Ein kompletter Programmierzyklus dauert nur ca. 1 Minute, ein Lesevorgang sogar nur etwa 2 Sekunden. Komplizierter ist die Bereitstellung der Programmierspannung von +25 V. Diese muß sehr konstant gehalten werden und darf auch unter stark schwankender Last (0 ... 30 mA) nicht die Grenzwerte des U 2716 C erreichen. Manche Import-EPROMs vertragen sogar noch weniger Toleranz der Programmierspannung (z. B. SU-Typ K 573 RF 2; +25 V $^{+0,5 V}_{-1 V}$). Es empfiehlt sich daher, die Programmierspannung in gewissen Grenzen variabel zu halten. Folgende Grundsätze sind zu beachten:

- Eine (auch nur durch überlagerte Impulse!) zu hohe Programmierspannung zerstört den EPROM unweigerlich!




Bild 2: Schaltbild des Programmiermoduls (R29 = 10 KOhm)

- Eine zu niedrige Programmierspannung gestattet, wenn überhaupt, nur eine oberflächliche Programmierung. Das bedeutet einen eventuellen Datenverlust im Laufe der Betriebszeit, schadet aber dem Bauelement ansonsten nicht.

Der Amateur wird aus Kostengründen die Programmierspannung nie an die obere Grenze legen, +24,5 V haben sich bei praktisch allen 2716-Typen als ausreichend erwiesen.

Neben der Einhaltung der Programmierspannungsgrenzen ist außerdem eine Strombegrenzung mit relativ scharfem Einsatz ratsam. Ein U 2716 C benötigt beim Programmieren aus der +25 V-Quelle einen Strom von maximal 30 mA. Wird eine Strombegrenzung auf diesen Wert vorgenommen, ist eine Zerstörung des EPROMs kaum noch zu befürchten. Auch der Schalttransistor T 3 wird damit geschont. Der praktische Aufbau des +25 V-Netzteiles kann auf verschiedene Weise erfolgen. Gut geeignet ist eine Schaltung mit dem integrierten Festspannungsregler MAA 723, da hier die Spannungskonstanz und die Strombegrenzung problemlos gewährleistet werden können.

Eine andere Möglichkeit bietet sich im Einsatz eines Transverters gemäß Bild 3 an.

Die Eingangsspannung ist eine Wechselspannung von 10 ... 12 V, die aus einem unabhängigen Netztrafo gewonnen wird (evtl.. zweites LC-80-Netzmodul). Der mit dem Timer-IC B 555 D bestückte Transverter bietet eine ausgezeichnete Spannungsregelung und über R 38 die Möglichkeit der Einstellung der Programmierspannung. Eine exakte Strombegrenzung ist nicht gewährleistet, jedoch ist die Stromabgabe des Transverters nicht sehr groß.

Ein Kurzschluß am Anschluß U_{PR} des U 2716 C sollte vermieden werden, da hierdurch zumindest T 5 gefährdet wird.

Vor der ersten Inbetriebnahme des Transverters ist das Potentiometer R 38 in die Endstellung zu drehen, die nicht an Masse liegt! Erst dann wird der Spannungssollwert eingeregelt.




Bild 3: Transverter für Programmierspannung (Der Schleifer des Potis R38, 4,7 KOhm, muß an der Basis von T4 angeschlossen sein, nicht am Kollektor. C 12 bleibt zwischen Basis und Kollektor)

4. Inbetriebnahme des Programmiermoduls

Alle Schaltungsteile werden auf richtige Verdrahtung und Bestückung geprüft, die Steckfassung auf dem Modul ist leer und der LC-80 abgeschaltet. Erst jetzt wird das Modul über den Steckverbinder X 1 mit dem LC-80 verbunden. Nach dem Anschluß der +25 V-Quelle an den Punkt P des Moduls und an Masse wird der LC-80 eingeschaltet.

Zuerst werden alle Pins der Programmierfassung auf Spannungsfreiheit kontrolliert. Kann an irgendeinem Pin nach dem Einschalten den LC-80 eine Spannung gegen Masse gemessen werden, liegt ein Fehler vor, der behoben werden muß!

Jetzt wird durch ein kurzes Programm der Anschluß B 0 der USER-PIO aus logisch "1" gebracht (siehe LC-80 - Handbuch, Seite 120 bzw. Seite 159 - ältere Ausführung -), B 1, B 2 und B 3 bleiben auf "0". Damit muß an den Pins 24 (U_{CC}) und 21 (U_{PR}) der Programmierfassung eine Spannung von ca. +5 V anliegen (an U_{PR} um die Diodendurchflußspannung von D 1 verringert). Dasselbe muß sich durch den B 1 -Anschluß der USER-PIO erreichen lassen.

Danach wird die Funktion der Signale /CE (durch B 2) und /OE (durch B 3) überprüft. Die Messung erfolgt an den Pins 18 bzw. 20 der Programmierfassung.

Am Pin 21 der Programmierfassung wird die Wirkung von B 3 auf U_{PR} geprüft. Sind B 1 und B 2 gleich 0, muß bei B 3 = 1 eine Spannung von +25 V zu messen sein, bei B 3 = 0 müssen 0 V anliegen. Wird dann B 1 auf 1 gesetzt, muß Pin 21 ca. +4,6 V führen.

Nach Abschluß dieser statischen Tests können die genannten Signale auch dynamisch kontrolliert werden, sofern ein Oszillograph zur Verfügung steht.

Zur optischen Kontrolle des Zustandes des Programmiermoduls im Betrieb dienen die beiden Leuchtdioden LED 1 und LED 2, die über entsprechende Beschaltung an den Pins 24 und 21 der Programmierfassung liegen. LED 1 signalisiert, daß an der

Fassung U_{cc} mit +5 V beschaltet ist, und LED 2, daß U_{PR} mit +25 V belegt ist. Damit wird die Funktionsfähigkeit dieser Schaltungsteile kontrollierbar und außerdem sichergestellt, daß bei EPROM-Wechsel alle Betriebsspannungen ausgeschaltet sind. Diese Forderung des Bauelementeherstellers ist unbedingt zu beachten!

5. Software

Das im folgenden dargestellte Softwarekonzept gestattet einen schrittweisen Aufbau. Durch ein in den RAM-Speicher eingegebenes Programm kann ein erster EPROM programmiert werden. Der Inhalt dieses EPROMs wird wieder dieses (entsprechend modifizierte) "Ladeprogramm" sein. Wird der so programmierte EPROM auf die ROM-Position 3 des LC-80 gebracht, sind alle weiteren EPROM-Programmierungen ohne RAM-Eingaben möglich; das Ladeprogramm gehört dann sozusagen zum Betriebssystem des LC-80. Voraussetzungen für dieses Konzept sind die unter 2. genannten Hardware-Umrüstungen des Lerncomputers.

Das Programmsystem ist vor allem als "Handwerkszeug" für eine erweiterte Nutzung des LC-80 vorgesehen, sein Hauptwert liegt also in seiner Anwendung. Deshalb wurde auf eine umfassende Beschreibung der Software verzichtet. Dagegen wird der praktische Umgang mit dem Programmsystem ausführlich erläutert.

Zur Vermeidung von Irrtümern wurde im Programmlisting auf die

sonst übliche Darstellung von Hezadezimalzahlen mit nachgestelltem "H" verzichtet. Dadurch findet der Benutzer alle Angaben so vor, wie sie auch auf dem Display und der Tastatur des LC-80 erscheinen.

5.1. Ladeprogramm im RAM

Das folgende Programmsystem besteht aus einem Hauptprogramm, mehreren Unterprogrammen und einer Tabelle für die Testdarstellung. Die Funktionen der einzelnen Programmteile werden kurz erläutert. Das Programmsystem beginnt auf der Adresse 2000 und endet bei 21A8. Der Bereich 2400 bis 2BFF (2 KByte-Zusatz-RAM) bleibt für die Daten reserviert, die in den EPROM geladen werden sollen.

5.1.1. Hauptprogramm

Das Hauptprogramm organisiert den gesamten Programmierablauf einschließlich der Funktionen:

- Leerkontrolle (Ist EPROM vollständig leer? - FF?)
- Einlesen eines programmierten EPROMs in den RAM-Bereich
- Vergleich Ist- und Sollwert jeder Zelle.

Die folgenden Programme gehen von der Bestückung des LC-80 mit einem U 2716 C aus. Bei Bestückung mit 2 ROMs U 505 D ändern sich alle mit * versehenen Adressen gemäß Bedienungsanleitung LC-80, S. 58!

2000	21 84 21	LD HL, 2184	} Eintragen von Start- und Zieladresse sowie Byte-Anzahl aus der Tabelle in die RAM-Plätze, die im Kassettenbetrieb File-Name usw. enthalten
2003	11 EB 23	LD DE, 23EB	
2006	01 06 00	LD BC, 0006	
2009	ED B0	LDIR	
200B	CD 4E 20	CALL INIT	Initialisierung Port B
200E	CD 64 21	H1: CALL AUS	Grundzustand: $V_{CC}, V_{PP} = 0 V$
2011	DD 21 90 21	LD IX, 2190	Anzeige "U 2716"
2015	CD 5A 04 *	CALL DAK 1	
2018	FE 1F	CP 1F	LD ?
201A	F5	PUSH AF	
201B	CC 58 20	CAZ LESEN	
201E	F1	POP AF	
201F	FE 1E	CP 1E	ST ?
2021	20 EB	JRNZ EB	H1
2023	CD 73 20	CALL LEER	
2026	06 05	LD B, 05	5 Programmierzyklen
2028	C5	H2: PUSH BC	
2029	CD 96 20	CALL PROGR	
2020	CD 5F 03	CALL TON	1 Ton aus dem Monitorprogramm
2027	CD D1 20	CALL VERGL	
2032	C1	POP BC	
2033	10 F3	DJNZ F3	H2
2035	F5	PUSH AF	
2036	CD 64 21	CALL AUS	
2039	F1	POP AF	
203A	38 09	JRC 09	H3
203C	DD 21 A5 05*	LD IX, 05A5	Anzeige "ERROR"
2040	CD 5A 04*	CALL DAK 1	
2043	18 C9	JR C9	H1
2045	DD 21 8F 21	H3:LD IX, 2187	Anzeige "READY"
2049	CD 5A 04*	CALL DAK 1	
204C	18 C0	JR C0	H1

5.1.2. Unterprogramm INIT

Das Unterprogramm INIT dient zur Initialisierung des Port B der USER-PIO.

```
204E  3E FF      LD A, FF      } PIO-Port B, Mode 3
2050  D3 FB      OUT FB      }
2052  3E F0      LD A, F0      } E/A-Definition:
2054  D3 FB      OUT FB      } B 0 ... B 3 = Ausgänge
                                   B 4 ... B 7 = Eingänge
2056  C9        RET
2057  00        NOP
```

5.1.3. Unterprogramm LESEN

Das Unterprogramm LESEN realisiert die Übernahme des Inhaltes des auf der Programmierfassung steckenden EPROMs U 2716 C in den RAM-Speicher des LC-80 (2. und 3. RAM, Adressen 2400 bis 2BFF). Nach Abschluß des Vorganges erscheint die Anzeige "READY".

```
2058  CD FD 20   CALL ADVOR
205B  CD 1C 21   L1: CALL ADINK
205E  CD 4F 21   CALL DATIN
2061  FD 77 00   LD (IY+00), A
2064  FD 23      INC IX
2066  CD 73 21   CALL ENDE
2069  20 F0      JRNZ F0      L1
206B  DD 21 8F 21 LD IX, 2187   Anzeige "READY"
206F  CD 5A 04*  CALL DAK 1
2072  C9        RET
```

5.1.4. Unterprogramm LEER

Das Unterprogramm LEER vergleicht einzeln jedes Byte des auf der Programmierfassung steckenden EPROMs mit FF. Sind alle Zellen leer, wird "LEER" angezeigt, ansonsten "ERROR".

2073	CD FD 20	CALL ADVOR	
2076	CD 1C 21	O1: CALL ADINK	
2079	CD 4P 21	CALL DATIN	
207C	FE FF	CP FF	Vergleich mit FF
207E	20 11	JRNZ 11	O2
2080	CD 73 21	CALL ENDE	
2083	20 F1	JRNZ F1	O1
2085	DD 21 96 21	LD IX, 2196	Anzeige "LEER"
2089	06 FF	LD B, FF	Anzeigetest
208B	CD 83 04*	O3: CALL DAK 2	
208E	10 FB	DJNZ FB	O3
2090	C9	RET	
2091	21 3C 20	O2: LD HL, 203C	
2094	E3	EXX (SP), HL	
2095	C9	RET	

5.1.5. Unterprogramm PROGR

Im Unterprogramm PROGR wird der Inhalt von RAM 2 und 3 byteweise in den auf der Programmierfassung steckenden EPROM geladen. Nach Abschluß erfolgt die Anzeige "READY". Pro Byte wird ein Programmierimpuls von ca. 10 ms erzeugt. Diese Zeit wird jedesmal durch Aufruf von DAK 2 realisiert, ein Nebeneffekt ist die Anzeige von Adressen und Daten auf dem Display.

2096	DD 21 F2 23	LD IX, 23F2	
209A	CD FD 20	CALL ADVOR	
209D	DB F9	IN F9	
209F	CB DF	SET 3, A	/OE => 1
20A1	CB 97	RES 2, A	/CE => 0
20A3	D3 F9	OUT F9	} Programmier- sperre
20A5	CD 1C 21	P1: CALL ADINK	
20A8	FD 7E 00	LD A, (IY+00)	
20AB	D3 F8	OUT F8	
20AD	CD C3 04*	CALL DADP	
20B0	21 F8 23	LD HL, 23F8	
20B3	5E	LD E, M	
20B4	23	INC HL	
20B5	56	LD D, M	
20E6	1B	DEC DE	
20B7	CD B7 04*	CALL ADRSDP	
20BA	DB F9	IN F9	
20BC	CB D7	SET 2, A	/CE => 1
20BE	D3 F9	OUT F9	} Programmier- impuls ca. 10 ms
20C0	CD 83 04	CALL DAK 2	
20C3	DB F9	IN F9	
20C5	CB 97	RES 2, A	/CE => 0
20C7	D3 F9	OUT F9	
20C9	FD 23	INC IY	
20CB	CD 73 21	CALL ENDE	
20CE	20 D5	JRNZ D5	P1
20D0	C9	RET	

5.1.6. Unterprogramm VERGL

Das Unterprogramm VERGL organisiert einen Bitmustervergleich zwischen EPROM und Zusatz-RAM. Bei Übereinstimmung erscheint "PASS", bei Nichtübereinstimmung "FAIL".

```
20D1  CD FD 20      CALL ADVOR
20D4  CD 1C 21     V1: CALL ADINK
20D7  CD 4F 21     CALL DATIN
20DA  FD 46 00     LD B, (IY+00)
20DD  FD 23        INC IY
20D7  B8           CP B
20E0  20 0C        JRNZ 0C          V2
20E2  CD 73 21     CALL ENDE
20E5  20 ED        JRNZ ED          V1
20E7  DD 21 89 21  LD IX, 2189  Anzeige "PASS"
20EB  37           SET CF
20EC  18 05        JR 05          V3
20EE  DD 21 A2 21  V2:LD IX, 21A2  Anzeige "FAIL"
20F2  B7           OR A          CY: = 0
20F3  06 FF        V3: LD B, FF  Anzeigezeit
20F5  75           V4: PUSH AF
20F6  CD 83 04*    CALL DAK 2
20F9  F1           POP AF
20FA  10 F9        DJNZ P9          V4
20FC  C9           RET
```

5.1.7. Unterprogramm ADVOR

ADVOR dient zur Herstellung der Anfangsbedingungen bei der Adressierung von RAM und EPROM sowie zur Nullsetzung des Byte-Zählers.

```
20FD  21 FE 23      LD HL, 23FE
2100  36 00        LD M, 00
2102  23          INC HL
2103  36 00        LD M, 00
2105  21 ED 23     LD HL, 23ED
2108  5E          LD E, M
2109  23          INC HL
210A  56          LD D, M
210B  D5          PUSH DE
210C  FD E1       POP IY
210E  23          INC HL
210F  11 F8 23    LD DE, 23F8
2112  01 02 00    LD BC, 0002
2115  ED B0       LDIR
2117  3E 07       LD A, 07
2119  D3 F9       OUT F9
211B  C9         RET
```

5.1.8. Unterprogramm ADINK

Das Unterprogramm ADINK organisiert die Erhöhung der Adressen von RAM und EPROM sowie des Byte-Zählers, bis der vorgegebene Betrag (2 KByte) erreicht wird.

2110	3E 0F	LD A, 0F	
211E	D3 FA	OUT FA	
2120	21 F8 23	LD HL, 23F8	
2123	DB F9	1: IN F9	Im Unterprogramm AUS wird zu dieser Marke gesprungen.
2125	CB 87	RES 0, A	
2127	D3 F9	OUT F9	
2129	7E	LD A, M	
212A	4F	LD C, A	
212B	D3 F8	OUT F8	
212D	23	INC HL	
212E	DB F9	IN F9	
2130	CB C7	SET 0, A	
2132	CB 8F	RES 1, A	
2134	D3 F9	OUT F9	
2136	7E	LD A, M	
2137	47	LD B, A	
2138	D3 F8	OUT F8	
213A	03	INC BC	
213B	70	LD M, B	
213C	2B	DEC HL	
213D	71	LD M, C	
213E	DB F9	IN F9	
2140	CB CF	SET 1, A	
2142	D3 F9	OUT F9	
2144	21 FE 23	LD HL 23FE	
2147	5E	LD E, M	
2148	23	INC HL	
2149	56	LD D, M	
214A	13	INC DE	
214E	72	LD M, D	
214C	2B	DEC HL	
214D	73	LD M, E	
214E	C9	RET	

5.1.9. Unterprogramm DATIN

Das Unterprogramm DATIN ist zuständig für das Einlesen eines EPROM-Bytes in das entsprechende RAM-Byte bei vorher organisierter Adressierung.

214F	DB F9	IN F9
2151	CB 97	RES 2, A
2153	D3 F9	OUT F9
2155	3E 4F	LD A, 4F
2157	D3 FA	OUT FA
2159	DB F8	IN F8
215B	F5	PUSH AF
215C	DB F9	IN F9
215E	CB D7	SET 2, A
2160	D3 F9	OUT F9
2162	F1	POP AF
2163	C9	RET

5.1.10. Unterprogramm AUS

Bei AUS wird auf allen PIO-Leitungen 0 ausgegeben, damit werden auch U_{CC} und U^{PR} abgeschaltet, ebenso /CE und /OE.

2164	3E 0F	LD A, 0F	
2166	D3 FA	OUT FA	
2168	21 8A 05 * ¹	LD HL, 058A	
216B	CD 23 21	CALL ADINK 1	Hier wird nicht ADINK angesprungen, sondern die Marke 1 in diesem Unterprogramm!
216E	3E 00	LD A, 00	
2170	D3 F9	OUT F9	
2172	C9	RET	

*¹) Bei Bestückung des LC-80 mit 2 ROMs U 505 D ist hier einzutragen:

2168	21 8A 09	LD HL, 098A	
------	----------	-------------	--

5.1.11. Unterprogramm ENDE

Das Unterprogramm ENDE vergleicht den Byte-Zähler mit 2 K (hexadezimal 800). Je nach Ergebnis wird das Zero-Flag gesetzt. Bei Z = 1 sind alle Bytes abgearbeitet.

2173	21 FE 23	LD HL, 23FE	
2176	4E	LD C, M	
2177	23	INC HL	
2178	46	LD B, M	
2179	21 EB 23	LD HL, 23EB	
217C	5E	LD E, M	
217D	23	INC HL	
217E	56	LD D, M	
217F	EB	EX DE, HL	
2180	87	OR A	CY = 0 setzen, Z-Flag setzen, wenn H1 = BC
2181	ED 42	SBC HL, BC	
2183	C9	RET	

5.1.12. Tabelle der Anzeigetexte und Ladeparameter

2184	00	}	Anzahl der zu übertragenden Bytes	
2185	08		(800 hexadezimal - 2048 dezimal - 2 KByte)	
2186	00		Anfangsadresse im RAM-Bereich (2400)	
2187	24			
2188	00	}	Anfangsadresse beim zu programmierenden	
2189	00		EPROM U 2716 C (0000)	
218A	AE	}	"S"	"PASS"
218B	AE		"S"	
218C	6F		"A"	
218D	4F		"P"	
218E	00			
218F	00			
2190	AB	}	"Y"	"READY"
2191	E9		"D"	
2192	6F		"A"	
2193	CE		"E"	
2194	48		"R"	
2195	00			
2196	00			
2197	48	}	"R"	"LEER"
2198	CE		"E"	
2199	CE		"E"	
219A	C2	"L"		
219B	00			
219C	EE	}	"6"	"U 2716"
219D	21		"1"	
2198	25		"7"	
219F	CD		"2"	
21A0	00			
21A1	E3	"U"		

21A2	00		
21A3	C2	"L"	} "FAIL"
21A4	21	"I"	
21A5	6F	"A"	
21A6	4E	"F"	
21A7	00		
21A8	FF		

Nach einer gründlichen Kontrolle des eingegebenen Programmes sollte dieses zunächst auf Kassette "gerettet" werden (Startadresse 2000, Endadresse 21A8, File-Name 2716).

Danach kann bereits, auch ohne EPROM, eine gewisse Funktionskontrolle von Hard- und Software erfolgen.

Dazu ist der LC-80 abzuschalten, das Programmiermodul anzustecken und alle Verbindungen herzustellen. Dann wird der Computer wieder eingeschaltet und das Kassettenprogramm "2716" eingelesen. Die EPROM-Programmierung bleibt leer.

- Tasten **RES** und **ADR** betätigen
- Spätestens jetzt müssen U_{CC} und U_{PR} auf 0 V-Potential liegen. LED 1 und LED 2 dürfen nicht leuchten.
- **EX** betätigen, es erfolgt die Anzeige "U 2716",
- U_{CC} und U_{PR} bleiben auf 0 V.
- **ST** betätigen, Anzeige "ERROR", U_{CC} und U_{PR} gehen auf +5 V. LED 1 muß leuchten.
- **EX** drücken, Anzeige "U 2716", U_{CC} und U_{PR} gehen wieder auf 0 V. LED 1 verlischt.
- **LD** betätigen, nach ca. 2 Sekunden erfolgt Anzeige "READY", während dieser Zeit müssen U_{CC} und U_{PR} auf +5 V liegen, LED 1 leuchtet.
- **EX** , Anzeige "U 2716"
- **RES** und **ADR**
- Jetzt Adresse 2400 eingeben, dort muß der Speicherinhalt 00 angezeigt werden. Dies gilt auch für alle Folgeadressen bis 2BFF.

Wenn alles wie beschrieben funktioniert hat, kann die Erprobung fortgesetzt werden. Durch Aus- und Wiedereinschalten des LC-80 wird der gesamte RAM-Speicher gelöscht und danach das Programm "2716" erneut geladen. Mit folgendem kurzen Programm wird es zusätzlich auf den Bereich 2400 ... 25A8 gebracht:

2200	21 00 20	LD HL, 2000	Anfangsadresse der Quelldaten
2203	11 00 24	LD DE, 2400	Anfangsadresse des Zielbereiches
2206	01 A8 01	LD BC, 01A8	Byte-Anzahl (01A8 hex.)
2209	ED B0	LDIR	
220B	76	HALT	

Das Programm wird auf Adresse 2200 gestartet und ist mit Aufleuchten der HALT-LED beendet. Ab Adresse 2400 muß jetzt ebenfalls das komplette Ladeprogramm stehen.

Da beabsichtigt ist, dieses Programm in den ersten zu programmierenden EPROM zu bringen, sind einige Korrekturen notwendig. Alle Unterprogramme und anderen Einsprungadressen sind so umzuschreiben, daß sie später auf der ROM-Position 3 erreicht werden können. Tabelle 3 zeigt die Adressen, die dort eingetragenen Daten und die neuen Daten, die statt dessen eingegeben werden müssen (siehe S. 30).

Auch diese Daten und Adressen werden nochmals überprüft. Hard- und Software sind nun bereit zur ersten Programmierung eines EPROMs. Dies geht dann wie folgt vor sich:

Adresse	alte Daten	neue Daten
2402	21	11
240D	20	10
2410	21	11
2414	21	11
241D	20	10
2425	20	10
242B	20	10
2431	20	10
2438	21	11
2448	21	11
245A	20	10
245D	21	11
2460	21	11
2468	21	11
246E	21	11
2475	20	10
2478	21	11
247E	21	11
2482	21	11
2488	21	11
2493	20	10
249C	20	10
24A7	21	11
24CD	21	11
24D3	20	10
24D6	21	11
24D9	21	11
24E4	21	11
24EA	21	11
24F1	21	11
256D	21	11

Tabelle 3

- Betätigen von `RES` , `ADR` und `EX` , es erscheint "U 2716".
- Ein (wirklich!) leerer EPROM U 2716 C wird (richtig herum!) in die Programmierfassung des Moduls gesteckt.
- `ST` wird gedrückt und nach kurzer Zeit (2 s) entweder "LEER" oder "ERROR" angezeigt: letzteres, wenn der EPROM nicht vollständig leer ist.
- Diese Anzeige erfolgt nur für wenige Sekunden, bei "LEER" beginnt sofort danach die Programmierung - man sieht dabei die jeweilige EPROM-Adresse und die dazugehörigen Daten für etwa 10 ms. Ist die letzte EPROM-Adresse (07FF) erreicht, erscheint "PASS" oder "FAIL" (bei nicht vollständiger richtiger Programmierung). Unabhängig davon, wie der Datenvergleich ausgeht, wird der gesamte Programmiervorgang noch viermal wiederholt. Dies ist notwendig, um die insgesamt vorgeschriebenen 50 ms Programmierzeit pro Adresse einzuhalten.
- Wenn wenigstens der letzte Programmierzyklus erfolgreich war, erscheint danach "PASS" und dann "READY".

Damit ist der erste EPROM auf dem Programmiermodul programmiert worden. Da er, wenn alles funktioniert hat, das modifizierte Ladeprogramm enthält, kann der LC-80 jetzt abgeschaltet werden.

5.2. Das Ladeprogramm im ROM 3

Der EPROM kann aus der Programmierfassung des Programmiermoduls genommen und in die auf Position 3 befindliche 24polige Steckfassung gesetzt werden. Das Programmiermodul kann dabei am USER-Bus angeschlossen bleiben.

Wird der LC-80 jetzt eingeschaltet, muß auf der Adresse 1000 das Ladeprogramm stehen, natürlich mit den schon geänderten Daten. Dieses Ladeprogramm beansprucht 01A8 Bytes (hex.), das ist nur ca. 1/4 der Kapazität des U 2716 C. Der noch freie Speicherplatz kann später für Erweiterungen des Betriebssystems oder andere Programme benutzt werden.

6. Bedienungsanleitung

Wenn zukünftig EPROMs gelesen oder beschrieben werden sollen, werden zuerst das Programmiermodul in den abgeschalteten LC-80 gesteckt und alle notwendigen Verbindungen hergestellt (insbesondere die Programmierspannung anschließen!). Erst dann dürfen der Computer und die 25 V-Quelle eingeschaltet werden!

- RES und ADR betätigen
- Adresse 1000 eingeben, EX drücken
- Anzeige "U 2716" erscheint

- Soll der Inhalt eines programmierten EPROMs (oder ROMs) in den RAM-Speicher den LC-80 ab Adresse 2400 gelesen werden, wird LD betätigt, nach Ausführung erscheint "READY" (ca. 2 s).
- Mit EX wird der Ausgangszustand wieder hergestellt: "U 2716".
- Soll ein EPROM geladen werden, müssen natürlich die dafür bestimmten Daten im RAM-Bereich ab 2400 stehen. Nach dem Drücken von ST erfolgt automatisch eine Leerkontrolle des zu programmierenden EPROMs. Ist der EPROM leer, erscheint kurz die Anzeige "LEER" und dann wird sofort programmiert. Ist er nicht vollständig leer, erscheint "ERROR" - es erfolgt dann keine Programmierung.
- Beim eigentlichen Programmiervorgang können Adressen und Daten beobachtet werden. Eine erfolgreiche Programmierung wird mit "PASS", ein Fehler mit "FAIL" quittiert. Nach insgesamt 5 Programmierzyklen wird "READY" angezeigt.

6.1. Teiloperationen

Bisher wurden nur komplette EPROMs gelesen oder geladen. Es ist aber möglich, auch kürzere Programme oder Einzelbytes in bereits teilweise programmierte EPROMs zu laden. Die Bedienung erfolgt dann wie folgt:

- RES, ST betätigen, die Anzeige ist dann " X.X.X.X.-F " (X bedeutet hier, daß die dort angezeigten Werte ohne Bedeutung sind).

- Über die Tastatur werden jetzt die zu übertragenden Bytes in hexadezimaler Form eingegeben. Bei 72 Bytes (hex.) ist die abschließende Anzeige dann " 0.0.7.2.-F "
- drücken, es erscheint die Anzeige " X.X.X.X.-S "
- Über die Tastatur wird die Startadresse im LC-80 eingegeben (meist im RAM-Bereich, vorzugsweise ab 2400; es ist aber genauso möglich, einen ROM-Bereich in einen EPROM zu übertragen - aber nicht umgekehrt!).
Nach der Eingabe z. B. der Startadresse 2000 ist die Anzeige dann " 2.0.0.0.-S "
- drücken, es erscheint die Anzeige " X.X.X.X.-E "
- Jetzt Eingabe der Startadresse im EPROM, der zu programmieren ist. Die EPROM-Adressen beginnen grundsätzlich bei 0000 und enden beim U 2716 C bei 07FF. Soll z. B. der EPROM-Bereich ab 01B0 beladen werden, muß auf dem Display zu lesen sein: " 0.1.B.0.-E "
- , betätigen
- , , , , also Adresse 100B eintragen!
- drücken es erscheint dann wieder die Anzeige "U 2716".
- Danach kann mit programmiert und mit gelesen werden, wie im Abschnitt 6. behandelt.

Zusammengefaßt:

Die Anfangsbedingungen zur Teilladung oder zum Teillesen eines EPROMs werden genauso gesetzt wie beim Betrieb des LC-80 mit Kas-
settenrecorder. Dabei haben die Anzeigen
aber eine andere Bedeutung:

F - Anzahl der zu behandelnden Bytes

S - Startadresse im LC-80 (z. B. RAM)

E - Startadresse im EPROM auf dem Pro-
grammiermodul

Im obigen Beispiel sollten 72 Bytes (F) aus dem RAM-Bereich ab 2000 (S) in die EPROM-Adressen ab 01B0 (E) geladen werden oder umgekehrt.

Nach einer solchen Manipulation darf das Ladeprogramm nur auf Adresse 100B gestartet werden, da sonst der Programmanfang

automatisch folgende Bedingungen setzen würde:

F = 0800
S = 2400
E = 0000.

6 2. Kombination verschiedener Unterprogramme

Unter 6,. Wurde die Handhabung des automatisierten Lade- bzw. Lesevorganges dargestellt, wobei der EPROM grundsätzlich komplett behandelt wurde. Unter 6.1. wurde gezeigt, wie dies mit EPROM-Teilbereichen erfolgt. In diesem Abschnitt wird gezeigt, wie die einzelnen Unterprogramme nach eigenen Wünschen zusammengestellt werden können. Voraussetzung dafür ist die erfolgte Festlegung der Anfangsbedingungen gemäß Abschnitt 6. (Manipulation mit dem gesamten EPROM) oder 6.1. (Behandlung nur eines Teilbereiches). Das Programm ist dann mit Adresse 1000 oder 100B zu starten, das Ergebnis ist "U 2716".

Nun werden RES und ADR betätigt und auf einem noch unbenutzten RAM-Bereich wird das folgende Programm eingegeben:

2000	CD 4E 10	OALI, INIT	
2003	CD XX 10	CALL XX	(gewünschtes Unterprogramm)
2006	CD 64 11	OALL AUS	siehe B. 36)
2009	76	HALT	

RES , ADR , EX

Das Unterprogramm XX kann dabei sein:

58 - LESEN
73 - LEERKONTROLLE
96 - PROGRAMMIEREN
D1 - VERGLEICHEN

Mit etwas Übung wird das Programmsystem relativ leicht beherrscht. Die konsequente Unterprogrammtechnik erlaubt es auch, später eigene Programmzusammenstellungen zu verwenden.

7. Löschung von EPROMs

Das ernsthafte Arbeiten mit Mikrorechnern setzt das Programmieren von EPROMs und damit auch deren vorschriftsmäßiges Löschen voraus. Beide Arbeiten werden im Gegensatz zu anderen elektronischen Bauelementen meist vom Anwender durchgeführt, der daher Mitverantwortung für die Qualität und Zuverlässigkeit des EPROMs trägt. Besonders das Löschen dieser Bauelemente setzt einige Kenntnisse über die Mechanismen ihrer Informationsspeicherung voraus. Die Informationsspeicherung bei EPROMs des Typs U 2716 C

beruht auf der Aufladung sog. Floating-Gates, die praktisch kleine Kondensatoren darstellen. Diese Kondensatoren werden mittels komplizierter Vorgänge beim Programmieren geladen und müssen dann ihre Ladung über Jahre (!) hinweg behalten. Die Ladungen auf den Floating-Gates steuern ihrerseits die Feldeffekttransistoren der Speichermatrix.

Nach außen haben diese Speicherzellen natürlich digitalen Charakter, d. h. ihre Ladungen bewirken einen Low- oder High-Pegel am entsprechenden Datenausgang. Intern kann es aber durch

- unzureichende Ladung
- Ladungsverluste oder
- unvollständiges Löschen

zu instabilen Zuständen kommen, d. h. ein programmiertes Bit geht verloren bzw. ein nicht programmiertes kommt hinzu. Es gibt auch Fälle, bei denen dies durch Temperatur- oder Spannungsschwankungen sporadisch auftritt.

Deshalb ist bei allen Arbeiten mit EPROMs auf die exakte Einhaltung von Vorschriften und Empfehlungen der Hersteller zu achten.

Beim Löschen gilt folgendes:

- Quarzglas- oder Keramikfenster vor dem Löschen säubern (Azeton, Spiritus), jede unnötige Verschmutzung vermeiden.
- EPROMs nach der latenten Löschzeit noch einmal mit der doppelten Zeit nachlöschen. Die latente Löschzeit ist die Zeit, nach der ein EPROM gerade auf allen Adressen leer ist (FF). Es empfiehlt sich, bei allen Exemplaren einmal diese Löschzeit zu ermitteln (z. B. Leerkontrolle im 5-Minuten-Abstand durchführen) und diese Zeit mit 3 multipliziert irgendwo zu vermerken. Dies ergibt dann zukünftig sichere Löschungen.

- Programmierte EPROMs lichtdicht verschließen, da auch Umgebungslicht (Tageslicht, Leuchtstofflampen) langfristig einen Ladungsverlust bewirken kann.

7.1. UV-Löschgerät zum Selbstbau

Voraussetzung für den Selbstbau sind Fachkenntnisse auf dem Gebiet der Elektroinstallation und die Einhaltung aller geltenden Bestimmungen des Gesundheits-, Arbeits- und Brandschutzes!

Die für die EPROM-Löschung notwendige Strahlung im Wellenlängenbereich 253 nm (UV-C-Strahlung) wird in geeigneter Form vom UV-Strahler HNS 8 geliefert. Dies ist eine leuchtstofflose Version der Leuchtstofflampe LS 8 (8 Watt) mit ähnlichen Abmessungen und elektrischen Anschlußbedingungen, jedoch UV-durchlässigem Glaskolben. Damit wird ein problemloser Geräteaufbau möglich, denn es sind alle Zubehörteile der LS 8 (Sockel, Vorschaltgerät LXGu 8, Starter St 20) zu verwenden.

Der Strahler HNS 8 wird in Dunstabzugshauben für Küchen zur Keimtötung benutzt und ist in Elektro- und Haushaltwarengeschäften für ca. 80,-- M zu haben.

Die Grundschialtung für den Betrieb der HNS 8 zeigt Bild 4.




Bild 4: Schaltschema für HNS 8-UV-Strahler

Praktisch muß der Aufbau so erfolgen, daß die HNS 8 in einem geschlossenen Gehäuse untergebracht ist und die zu löschenden EPROMs zur Oberfläche des Strahlers einen Abstand von ca. 25 mm haben. Eine Zeitautomatik (z. B. Kurzzeitwecker oder elektronische Abschaltung z. B. mit E 355 D) sollte einen unbeabsichtigten Dauerbetrieb unbedingt verhindern (siehe auch Punkt 7.2.!).

7.2. Gesundheits-, Arbeits- und Brandschutz

Das EPROM-Löschgerät ist beim Bau und im Betrieb nicht ganz ungefährlich. Deshalb einige wichtige Hinweise:

Aufbau

- Das Gerät ist netzbetrieben! Damit gelten alle Bestimmungen der TGL 200-602 und TGL 200-611. Insbesondere die erforderlichen Schutzmaßnahmen zur Verhinderung der Berührung spannungsführender Teile sind peinlich einzuhalten. Empfohlen wird der Aufbau in Schutzkontaktausführung (Schuko) mit Blechgehäuse und Schutzleiteranschluß, da ein Kunststoffgehäuse (Voraussetzung für die Ausführung in Schutzisolierung) infolge der ständigen UV-Bestrahlung sehr schnell altert und die mechanische Festigkeit verliert. Die Schutzkontaktausführung verlangt natürlich den Betrieb ausschließlich an Schutzkontaktsteckdosen!
- Es ist unbedingt darauf zu achten, daß Augen und Haut nicht der UV-Strahlung ausgesetzt werden. Schon wenige Sekunden Einwirkungszeit auf die Augen können schwere Verletzungen verursachen, da das UV-Licht der HNS 8 viel kurzwelliger ist als z. B. das der Höhensonne! Deshalb sind Funktionsprüfungen in der Aufbauphase grundsätzlich mit der elektrisch identischen Leuchtstoffröhre LS 8 durchzuführen.
- Für den Betrieb mit dem UV-Strahler HNS 8 ist unbedingt eine vollständige Abdeckung des Strahlers (auch gegen Streulicht) zu gewährleisten. Konstruktiv kann das z. B. durch einen Einschub gelöst werden, auf dem die EPROMs in das Gerät eingeschoben werden.
Eine andere Möglichkeit ist eine Konstruktion mit Deckel, der erst beim Schießen über Mikrotaster den UV-Strahler

mit dem Netz koppelt (auf Berührungssicherheit achten!).
Insgesamt muß die Konstruktion und die Ausführung "narrensicher" sein - insbesondere bei Anwendung im Hobbybereich (Kinder !!).

Betrieb

Zusätzlich zu den o. a. Gefahren kommen beim Betrieb noch weitere hinzu.

Die extrem kurzweilige Strahlung erzeugt gefährliche Schadstoffe, z. B. Ozon und nitrose Gase. Zwar entstehen nur geringe Konzentrationen, ein Dauerbetrieb ist aber auf jeden Fall zu vermeiden. Deshalb sollten die Löschvorgänge nicht allzu häufig erfolgen und dann gleich mehrere EPROMs gleichzeitig behandelt werden.

Folgende Regeln sind einzuhalten:

1. Nur Kurzzeitbetrieb (ca. 30 Minuten) durchführen!
2. Für ausreichende Belüftung sorgen!
3. Bestrahlung (auch kurzzeitig) von Haut und Augen ausschließen!
4. Unkontrollierten Betrieb vermeiden! Keinesfalls Abschalten vergessen!
5. Gerät nur an Steckdose mit Schutzkontakt betreiben!
6. Gerät nach dem Abschalten noch ca. 10 Minuten geschlossen lassen, dadurch erfolgt ein Abbau der im Inneren befindlichen Gase!

8. Literaturverzeichnis

1. Bedienungsanleitung Lerncomputer LC-80 1.
Ausgabe November 1984
VEB Mikroelektronik "Karl Marx" Erfurt - Stammbetrieb
2. Handbuch LC-80
VEB Mikroelektronik "Karl Marx" Erfurt - Stammbetrieb
3. Hertzsch, A.: CMOS-Logikschaltkreise
Band 212 aus der Reihe "elektronika"
Militärverlag der DDR, Berlin 1983
4. Schultze, K./Trettin, T./Henke, K.:
Programmieren der EPROMs U 555 C und U 2716 C
radio fernsehen elektronik, Jahrgang 34 (1985), Heft 11,
Seite 700
Verlag Technik
5. TGL 43077/01 - 04
Unipolarer UV-löschbarer Festwertspeicherschaltkreis U
2716 C

RFT



**veb mikroelektronik
karl marx erfurt
stammbetrieb**

DDR-5023 Erfurt, Rudolfstraße 47
Telefon: 5 80, Telex: 061306

**elektronik
export·import**

Volkseigener Außenhandelsbetrieb der
Deutschen Demokratischen Republik
DDR - 1026 Berlin, Alexanderplatz 6
Telex: BLN 114721 elei, Telefon: 2180